
Abstract
This paper presents a construction of the real algebraic numbers
with executable arithmetic operations in Isabelle/HOL. Instead of
verified resultants, arithmetic operations on real algebraic numbers
are based on a decision procedure to decide the sign of a bivariate
polynomial (with rational coefficients) at a real algebraic point. The
modular design allows the safe use of fast external code. This work
can be the basis for decision procedures that rely on real algebraic
numbers.

Categories and Subject Descriptors I.1.1 [Symbolic and Al-
gebraic Manipulation]: Expressions and Their Representation—
Representations (general and polynomial); D.2.4 [Software En-
gineering]: Software/Program Verification—Correctness proofs,
formal methods

Keywords Real algebraic geometry, Isabelle/HOL, Decision pro-
cedure

1. Introduction
Real algebraic numbers (e.g.

√
2 or 3 + 2

√
5) are real numbers that

are defined as particular roots of non-zero polynomials with rational
(or integer) coefficients. They are important in computer algebra as
each one can be encoded precisely (unlike real numbers), and their
arithmetic and comparison operations are decidable. Formalizing
them in Isabelle/HOL [13], an interactive theorem prover, opens
the way to numerous decision procedures in computer algebra. For
example, consider this real closed problem in existential form:

∃x y ∈ R. x2 − 2 = 0 ∧ xy > 1

With our formalization of executable arithmetic and comparison
operations, we can prove it computationally: we search for a real
algebraic point (e.g. through cylindrical algebraic decomposition
[1]) that satisfies the quantifier-free part of the formula. One such
solution is (x =

√
2, y = 1).

Our formalization1 follows Isabelle’s tradition of separation of
abstraction and implementation. That is,

1 Source is available from https://bitbucket.org/liwenda1990/
real_algebraic_numbers

• abstraction: we first formalize real algebraic numbers on an
abstract level without considering executability (see Section 2).
More specifically, we formalize real algebraic numbers as a
subtype of real numbers, and show them to form an ordered field
using classic proofs in abstract algebra.
• implementation (restoring executability): we then restore ex-

ecutability on real algebraic numbers (see Section 3). More
specifically, we define a pseudo constructor for algebraic real
numbers and prove code equations for algebraic arithmetic and
comparison on this constructor. Some of the code equations
for algebraic arithmetic are based on a verified decision proce-
dure to decide the sign of a bivariate polynomial with rational
coefficients at real algebraic points.

The whole project is intended for practical applications. We aim
to build decision procedures related to real algebraic numbers on
the top of our current formalization.

The paper continues as follows. The first component of our
modular design is the abstract specification of real algebraic numbers
(§2), which is then followed by an implementation (§3) in the form
of Isabelle/HOL code equations. In particular, the implementation
is concerned with deciding the signs of polynomials at a given real
algebraic point. Related work (§5) is then described along with a
discussion of various limitations of and extensions to the work (6),
followed by conclusions (§7).

2. Construction on an Abstract Level
This section presents our formalization of real algebraic numbers as
an abstract data type. Definitions on this level will be as abstract as
possible without considering executability.

Mathematically, a real algebraic number α is a real number
for which there exists a non-zero univariate polynomial P (x) with
integer (or rational) coefficients, such that P (x) = 0 when x = α.

It is then straightforward to define the real algebraic numbers
as a subset of the real numbers. We formalise this construction by
defining type alg as a subtype2 of type real :

typedef alg = "{x::real. ∃ p::int poly. p 6=0
∧ poly (of_int_poly p) x = 0}"

where of_int_poly converts coefficients of a polynomial from int
to real, and poly p x means evaluating polynomial p at x.

To prove non-trivial properties about real algebraic numbers, we
need at least to prove that they are closed under the basic arithmetic
operations and hence form a field. For example, to show that real
algebraic numbers are closed under addition, suppose we have two
real algebraic numbers α and β, given by polynomials P and Q:

α ∈ R, P ∈ Z[x] P 6= 0 ∧ P (α) = 0

β ∈ R, Q ∈ Z[x] Q 6= 0 ∧Q(β) = 0

2 A description of Isabelle/HOL subtype definitions can be found in the
Tutorial [13, §8.5.2]

Then we have to show that

∃R ∈ Z[x]. R 6= 0 ∧R(α+ β) = 0. (1)

One way to show this is to compute R constructively using
resultants as in Cyril Cohen’s proof in Coq [2]. However, as we are
working on an abstract level and not concerned with executability,
a non-constructive but usually simpler proof (to show the mere
existence of such a polynomial) seems more appealing. Therefore,
we decided to follow a classic proof in abstract algebra.

Definition 1 (vector space). A vector space V over a field F is
an abelian group associated with scalar multiplication αv for all
α ∈ F and v ∈ V , satisfying the standard additivity and identity
axioms.

In the Multivariate_Analysis library in Isabelle/HOL, the notion
of a vector space is formalized using a locale :

locale vector_space =
fixes scale :: "’a::field ⇒ ’b::ab_group_add ⇒ ’b"
assumes "scale a (x + y) = scale a x + scale a y"
and "scale (a + b) x = scale a x + scale b x"
and "scale a (scale b x) = scale (a * b) x"
and "scale 1 x = x"

where scale::’a ⇒’b ⇒’b denotes scalar multiplication for ’a
a field and ’b an abelian group.

The standard library development of vector spaces has been ex-
tended by Jose Divasón and Jesús Aransay in their formalization of
the Rank-Nullity Theorem in Linear Algebra, including definitions
of span and of linearly dependent [4].

Definition 2 (Span). Let S = {v1, v2, ..., vn} be a set of vectors in
a vector space, then span(S) is defined as

{w | w = a1v1 + a2v2 + ...+ anvn, and a1,...,an are scalars}
Divasón and Aransay [4] formalize span slightly differently, but

the following lemma can be considered as an alternative definition
that matches standard mathematical definitions:

lemma span_explicit:
"span P = {y. ∃ S u. finite S ∧ S ⊆ P

∧ setsum (λv. scale (u v) v) S = y}"

where u of type ’b ⇒’a maps each vector in S to the corresponding
scalar. And setsum (λv. scale (u v) v) S maps each element
in S using (λv. scale (u v) v) and sums the results.

Definition 3 (linearly dependent). Let S = {v1, v2, ..., vn} be a
set of vectors in a vector space, we say S is linearly dependent if
there exist scalars a1, a2, ..., an, at least one of which is non-zero,
such that

a1v1 + a2v2 + ...+ anvn = 0

Divasón and Aransay [4] formalize dependent as

definition "dependent S ←→ (∃ a ∈ S. a ∈ span (S - {a}))"

since

a1v1 + a2v2 + ...+ anvn = 0

⇐⇒ vn =
a1
−an

v1 +
a2
−an

v2 + ...+
an−1

−an
vn−1

assuming an is the non-zero scalar.

Now, back to the problem of showing the formula (1), we can
consider the vector space of reals with rational scalars:

interpretation rat: vector_space
"(λx y. (of_rat x * y))::rat ⇒ real ⇒ real"

where of_rat :: rat ⇒ real embeds rat into real and the
scale function in vector_space is instantiated as

(λx y. (of_rat x * y))::rat ⇒ real ⇒ real

After the interpretation, we have new constants, such as

rat.span :: real set ⇒ real set and
rat.dependent :: real set ⇒ bool

that instantiate constants such as vector_space.span and
vector_space.dependent, and inherit all associated lemmas from
vector_space.

If we can show that {1, x, x2, ..., xn} is linearly dependent, then
(by the definition of linear dependence) it is not hard to see that
there exists a non-zero polynomial with rational coefficients and
degree at most n that vanishes at x:

lemma dependent_imp_poly:
fixes x::real and n::nat
assumes "rat.dependent {x ^ k |k. k ≤ n}"
shows "∃ p::rat poly. p 6= 0 ∧ degree p ≤ n ∧

poly p x = 0"

Now the problem becomes, how can we deduce the linear
dependence of a set of vectors? The solution is based on a lemma:
if m vectors live in the span of n vectors with m > n, then these m
vectors are linearly dependent.

lemma ( in vector_space) span_card_imp_dependent:
fixes S B::"’b set"
assumes "S ⊆ span B" and "finite B"

and "card S > card B"
shows "dependent S"

Moreover, we can also show for all n ∈ N

(α+β)n ∈ span{αiβj | i j ∈ N.i ≤ deg(P )∧j ≤ deg(Q)}
which can be derived by the following lemma in Isabelle:

lemma bpoly_in_rat_span:
fixes p q::"rat poly" and x y::real

and bp::"rat poly poly"
assumes "poly p x = 0" and "p 6= 0"
assumes "poly q y = 0" and "q 6= 0"
shows "bpoly bp x y ∈ rat.span {x ^ k1 * y ^ k2

|k1 k2. k1 ≤ degree p ∧ k2 ≤ degree q}"

Above, bp::"rat poly poly" means a bivariate polynomial with
rational coefficients and bpoly bp x y evaluates bp at (x, y). It
follows that

1, (α+ β), ..., (α+ β)(deg(P )+1)(deg(Q)+1)

are linearly dependent by applying Lemma span_card_imp_dependent,3

since

(deg(P ) + 1)(deg(Q) + 1) + 1

> card{αiβj | i j ∈ N.i ≤ deg(P ) ∧ j ≤ deg(Q)}
= (deg(P ) + 1)(deg(Q) + 1)

hence there exists a non-zero polynomial with integer coefficients4

vanishing at α+ β. Similarly, there exist such polynomials for the
difference α− β and the product αβ:

lemma root_exist:
fixes x y::real and p q::"rat poly"
assumes "poly p x = 0" and "p 6=0"
assumes "poly q y = 0" and "q 6=0"
defines "rt≡(λz::real. ∃ r::int poly. r 6=0
∧ poly r z = 0)"

shows "rt (x+y)" and "rt (x-y)" and "rt (x*y)"

3 In fact, there are corner cases when α+ β = −1, 0, 1, but all of them can
be satisfied, so the conclusion holds.
4 We have a lemma to convert a polynomial with rational coefficients into
one with integer coefficients, multiplying out the denominators.

Every rational number r is real algebraic, given by the root of the
first degree polynomial x − r. Therefore 0 − α is real algebraic,
covering the case of −α.

As for the multiplicative inverse, let

P (x) = anx
n + an−1x

n−1 + ...+ a0.

Then clearly

P (α) = 0 ∧ α 6= 0

=⇒ a0

(
1

α

)n

+ an−1

(
1

α

)n−1

+ ...+ an = 0 (2)

and hence we get vanishing polynomials for 1/α:

lemma inverse_root_exist:
fixes x::real and p::"rat poly"
assumes "poly p x = 0" and "p 6=0"
shows "∃ q::int poly. q 6=0 ∧ poly q (inverse x) = 0"

as well as α/β (treated as α× (1/β)).
Finally, to define arithmetic operations on alg, we lift the corre-

sponding operations from its underlying type, real. For example,
addition on alg is defined as

lift_definition plus_alg :: "alg ⇒ alg ⇒ alg"
is "plus::real ⇒ real ⇒ real"

which leaves us a goal to show that the invariant condition on alg is
maintained (that alg is closed under addition):∧
r1 r2 ::real.
∃ p. p 6= 0 ∧ poly (of_int_poly p) r1 = 0 =⇒
∃ p. p 6= 0 ∧ poly (of_int_poly p) r2 = 0 =⇒
∃ p. p 6= 0 ∧ poly (of_int_poly p) (r1 + r2) = 0

and this goal can be discharged by our previous Lemma root_exist.
Similarly, we obtain 0::alg and 1::alg, and the ordering operations
are lifted from real as well:

lift_definition zero_alg::alg is "0::real"

lift_definition one_alg::alg is "1::real"

lift_definition less_alg::"alg ⇒ alg ⇒ bool"
is "less::real ⇒ real ⇒ bool"

The command lift_definition is part of Isabelle’s Lifting and Transfer
package [9].

With zero, one, arithmetic and ordering operations defined, it
follows that alg forms an ordered field:

instantiation alg :: linordered_field

Because alg is essentially a subtype of real, all the instance
proofs of the instantiation above are one-liners, again thanks to the
Lifting and Transfer package [9]. For example, the associativity of
alg multiplication is proved by the following tactic:

show "(a * b) * c = a * (b * c)"
by transfer auto

And so, we have constructed the real algebraic numbers on an
abstract level and proved that they form an ordered field. But now it
is time to consider the question of executability.

3. Implementation
Executability is a key property of real algebraic numbers. They
are a countable subset of the real numbers and can be represented
exactly in computers. This section will demonstrate how we have
implemented algebraic real numbers and achieved executability on
their arithmetic operations through verified bivariate sign tests.

3.1 More Pseudo-Constructors on Real Numbers
Recall that our alg is actually a subtype of real, hence executability
on real operations can be reflected in alg. Therefore, our following
focus is to extend executability on type real.

The set of real numbers, as we know, is uncountable, hence
not every real number can be encoded finitely. That is, arithmetic
operations can only be executable on a strict subset of the real
numbers. Prior to our work, arithmetic operations on type real in
Isabelle were only executable on rational numbers embedded into
the reals (rational reals). For example, the following expression
could be evaluated to be true:

value "Ratreal (3/4) * Ratreal 2 > (0::real)"

where Ratreal of type rat ⇒ real is a pseudo-constructor [6]
that constructs a real from a rat. With Ratreal, code equations
such as

lemma real_plus_code [code]:
"Ratreal x + Ratreal y = Ratreal (x + y)"

restore executability on rational reals by mapping arithmetic opera-
tions on rational reals to rational numbers.

Our formalization follows a similar approach. We want to have
a constructor Alg of type _ ⇒ real to construct (algebraic) reals
from some encodings of real algebraic numbers.

An encoding (of a real algebraic number) is essentially a poly-
nomial (with integer or rational coefficients) and a root selection
strategy to distinguish a particular real root of the polynomial from
any others. There are several such strategies, such as using a ratio-
nal interval that only includes the target root, a natural number to
indicate the index of the root and Thom encoding [1, p. 42]. We
have decided to use the interval strategy, which is straightforward
to implement. Therefore, Alg is of type int poly ⇒ rat ⇒ rat
⇒ real, where the two rat arguments represent an interval.

As each real number in Isabelle is presented as a Cauchy se-
quence of type nat ⇒ rat, we explicitly construct such a sequence
using a suitable encoding:5

fun to_cauchy:: "rat poly × rat × rat ⇒ nat ⇒ rat"
where

"to_cauchy (_, lb, ub) 0 = (lb+ub)/2"
| "to_cauchy (p, lb, ub) (Suc n) = (

let c = (lb+ub)/2
in if poly p lb * poly p c ≤ 0

then to_cauchy (p, lb, c) n
else to_cauchy (p, c, ub) n)"

The idea is to bisect the given interval at each stage. The midpoint c
is determined as the average of the lower and upper bounds. Recall
that the polynomial has exactly one root in this interval. If this root
lies in the first half (indicated by a change of sign in the polynomial)
then this half is chosen, and otherwise the opposite half.

We can now define the constructor Alg :

definition Alg:: "int poly ⇒ rat ⇒ rat ⇒ real" where
"Alg p lb ub = (if valid_alg p lb ub

then Real (to_cauchy (p,lb,ub))
else undefined)"

where valid_alg p lb ub checks several validity conditions,

• lb < ub,
• poly p lb * poly p ub <0

• p contains exactly one real root6 within the interval (lb, ub)

5 An int poly can trivially be mapped to a rat poly.
6 This can be computationally checked using Sturm’s theorem, which is
a special case of our previously formalized Sturm-Tarski theorem (see
Theorem 1 in the next section).

• the interval (lb,ub) excludes 0 unless 0 is a root of p

The function Real, of type (nat ⇒ rat) ⇒ real, is the abstrac-
tion function that constructs a real number from its representation
as a Cauchy sequence.

Some useful lemmas regarding reals constructed by Alg can be
derived. For example, provided valid_alg p lb ub, we can show
that Alg p lb ub lies within the interval (lb, ub) and is a root of the
polynomial p:

lemma alg_bound_and_root:
fixes p::"int poly" and lb ub::rat
assumes "valid_alg p lb ub"
shows "lb < Alg p lb ub" and "Alg p lb ub < ub"

and "poly p (Alg p lb ub) = 0"

Note that we have described the constructor Alg previously [11]
and repeat it here for completeness.

3.2 Deciding the Sign of a Bivariate Polynomial at a Real
Algebraic Point

We have previously formalized the Sturm-Tarski theorem, and used
it to decide the sign of any univariate polynomial with rational
coefficients at a real algebraic point [10, 11]. In this section, we shall
formalize a sign determination algorithm for bivariate polynomials.
Note that R denotes R∪{−∞,∞}, the reals extended with infinity.

Definition 4 (Tarski Query). The Tarski query TaQ(Q,P, a, b) is

TaQ(Q,P, a, b) =
∑

x∈(a,b),P (x)=0

sgn(Q(x))

where a, b ∈ R, P,Q ∈ R[X], P 6= 0 and sgn : R → {−1, 0, 1}
is the sign function.

Essentially, the Sturm-Tarski theorem (sometimes known simply
as Tarski’s theorem [1]) provides a way to compute Tarski Queries
using some remainder sequences:

Theorem 1 (Sturm-Tarski). Every Tarski query satisfies

TaQ(Q,P, a, b) = Var(SPRemS(P, P ′Q); a, b),

where P 6= 0, P,Q ∈ R[X], P ′ is the first derivative of P , a, b ∈ R,
a < b and are not roots of P . Moreover, SPRemS is the signed
pseudo remainder sequence and

Var([p0, p1, ..., pn]; a, b)

= Var([p0(a), p1(a), ..., pn(a)])−Var([p0(b), p1(b), ..., pn(b)])
is the difference in the number of sign variations (after removing
zeroes) in the polynomial sequence [p0, p1, ..., pn] evaluated at a
and b.

Note that previously [10, 11] we have used the signed remainder
sequence SRemS, in which the remainder operation (mod) is from
Euclidean division on polynomials:

P = (P divQ) Q+ (P modQ)

and if Q 6= 0 thendeg(P modQ) < deg(Q).

Here, the signed pseudo remainder sequence SPRemS is based on
polynomial pseudo-divisions (pmod and pdiv):

lc(Q)1+deg(P )−deg(Q)P = (P pdivQ) Q+ (P pmodQ)

and if Q 6= 0 thendeg(P pmodQ) < deg(Q)

and lc(Q) is the leading coefficient of Q (3)

The key difference between Euclidean divisions and pseudo-
divisions is that Euclidean divisions can only be carried out on
polynomials over a field while pseudo-divisions are suitable for
polynomials over an integral domain.

The signed pseudo remainder sequence SPRemS is imple-
mented as follows:

function spmods :: "’a::idom poly ⇒ ’a poly
⇒ (’a poly) list" where

"spmods p q= (if p=0 then [] else
let

m=(if even(degree p+1-degree q)
then -1
else -lead_coeff q)

in
Cons p (spmods q (smult m (p pmod q))))"

where lead_coeff p is the leading coefficient of the polynomial p,
pmod is the pseudo remainder operation satisfying Equation (3), and
smult of type ’a ⇒’a poly ⇒’a poly is scalar multiplication
that multiplies a polynomial by a constant.

Here is an example to decide the sign of α − 1 at α =
√
2 =

(x2 − 2, 1, 2):

TaQ(x− 1, x2 − 2, 1, 2)

= Var(SPRemS(x2 − 2, (2x)(x− 1)); 1, 2)

= Var([x2 − 2, 2x2 − 2x, 8− 4x,−64]; 1, 2)
= 2− 1 = 1

As TaQ(x− 1, x2− 2, 1, 2) = 1 and
√
2 is the only root of x2− 2

within (1, 2), we can say that α− 1 is positive at α =
√
2.

To illustrate our idea for a bivariate sign determination procedure,
suppose we want to decide the sign of Q(x, y) ∈ Q[x, y] at (α, β)
with α = (P1, a1, b1) and β = (P2, a2, b2). By substituting y by β,
we haveQ(x, β) as a univariate polynomial in Q(β)[x], where Q(β)
is the field Q extended by β. Pretending to have arithmetic of real
algebraic numbers, we can still use the univariate sign determination
procedure:

TaQ(Q(x, β), P1(x), a1, b1)

= Var(SPRemS(P1(x), P1(x)
′Q(x, β)); a1, b1) (4)

To proceed from (4), we need to somehow eliminate algebraic
arithmetic in the operation pmod inside SPRemS. A key lemma is

lemma poly_y_dist_pmod:
fixes p::"’a::idom poly poly" and y::’a
assumes "poly (lead_coeff p) y 6=0"

and "poly (lead_coeff q) y 6=0"
shows "(poly_y p y) pmod (poly_y q y) =

poly_y (p pmod q) y"

where ’a poly poly is the type we use to represent bivariate polyno-
mials in Isabelle/HOL. This is the so-called recursive representation,
where for example, the bivariate polynomial

4xy + 3x+ 2y + 1 = 1 + 2y + (3 + 4y)x ∈ (Z[y])[x]

is encoded as [:[:1,2:],[:3,4:]:]. Moreover, the function poly_y
p a substitutes the value a for variable y in p. For example,

value "poly_y [:[:1,2:],[:3,4:]:] (2::int)"

evaluates to [:5, 11:], which can be mathematically interpreted as
(4xy + 3x+ 2y + 1)[y → 2] = 5 + 11x.

An important point about Lemma poly_y_dist_pmod is that the
left-hand occurrence pmod operates over Q(β)[x] (as poly_y p y
can be considered to be of type Q(β)[x], provided p ∈ Q[x, y] and
y is instantiated to β), which demands algebraic arithmetic, while
the right-hand occurrence of pmod operates over Q[x, y], which only
requires arithmetic over rational numbers. Therefore, provided the
leading coefficients of p and q do not vanish when evaluating at
y, i.e. poly (lead_coeff p) y 6= 0 and poly (lead_coeff q) y
6= 0 ), we can eliminate algebraic arithmetic in pmod.

In order to rewrite with Lemma poly_y_dist_pmod inside a
remainder sequence, we need to satisfy its assumptions. Therefore,
we have defined a function degen (for ‘degenerates’) of type ’a
poly poly ⇒’a ⇒’a poly poly, such that degen p y iteratively
removes the leading coefficient of p until it does not vanish at y or
p becomes 0 :

lift_definition degen::"’a poly poly ⇒ ’a ⇒’a poly poly" is
"λp y n. (if poly_y p y 6=0 ∧ n ≤degree (poly_y p y)

then coeff p n else 0)"

Note that the term (λp y n. ...) above is of type ’a poly poly
⇒ ’a ⇒ nat ⇒ ’a poly, so degen is defined in a way where
degen p y (of type ’a poly poly ) is lifted from its underlying
representation,7 which is of type nat ⇒’a poly.

For example, a bivariate polynomial 1 + y + (y2 − 2)x2

degenerates to 1 + y when y =
√
2, hence the command

value "degen[:[:1,1:],0,[:-2,0,1:]:] (Alg [:-2,0,1:] 1 2)"

evaluates to [:[:1,1:]:].

Properties of degen include that degenerating the bivariate
polynomial p with respect to y does not affect the result of evaluating
it at y :

lemma poly_y_degen: "poly_y (degen p y) y = poly_y p y"

This holds because only leading coefficients that vanish at y are
removed. Moreover, the leading coefficient of degen p y will not
vanish at y unless p vanishes at y :

lemma degen_lc_not_vanish:
assumes "degen p y 6=0"
shows "poly (lead_coeff (degen p y)) y 6=0"

With the help of degen, we can define another remainder se-
quence spmods_y that is similar to the previous signed pseudo re-
mainder sequence spmods except for that spmods_y p q y keeps
degenerating each remainder with repect to y :

function spmods_y :: "’a::idom poly poly ⇒ ’a poly poly
⇒’a ⇒ (’a poly poly) list" where

"spmods_y p q y= (if p=0 then [] else
let

mul=(if even(degree p+1-degree q)
then -1
else -lead_coeff q);

r= degen (smult mul (p pmod q)) y
in

Cons p (spmods_y q r y))"

By exploiting Lemma poly_y_dist_pmod, we have established
the relationship between spmods and spmods_y :

lemma spmods_poly_y_dist:
fixes p q :: "’a::idom poly poly"

and y::"’a::idom"
assumes "poly (lead_coeff p) y 6=0"

and "poly (lead_coeff q) y 6=0"
shows "spmods (poly_y p y) (poly_y q y)

= map (λp. poly_y p y) (spmods_y p q y)"

Note, similar to what we have stated for Lemma poly_y_dist_pmod,
the importance of Lemma spmods_poly_y_dist is that the left-
hand remainder sequence (spmods ) requires arithmetic over Q(β)[x]
(provided p q ∈ Q[x, y] and y = β) while the right-hand sequence
(spmods_y ) only requires arithmetic over Q[x, y].

7 ’a poly is constructed as a subtype of nat ⇒’a (i.e. a mapping from
exponents to coefficients). Haftmann et al. [7] discuss how to formalize
polynomials in Isabelle/HOL.

Let spmods_y p q y represented as SPRemS′(p, q, y), we can
rewrite SPRemS with Lemma spmods_poly_y_dist :

lcx(Q)(β) 6= 0 =⇒
SPRemS(P1(x), P1(x)

′Q(x, β)) =

SPRemS′(P1(x), P1(x)
′Q(x, y), β)[y → β] (5)

where lcx(Q) ∈ Q[y] is the leading coefficient of the bivariate
polynomial Q ∈ Q[x, y] with respect to x. [y → β] performs
substitution on a list of polynomials. For example, let [x, x+ y] be
a list of polynomials, then [x, x+ y][y → 3] = [x, x+ 3].

By equations (4) and (5), we have

lcx(Q)(β) 6= 0 =⇒
TaQ(Q(x, β), P1(x), a1, b1)

= Var(SPRemS′(P1(x), P1(x)
′Q(x, y), β)[y → β]; a1, b1)

(6)

Note, SPRemS′ operates over Q[x, y] and Var requires deciding
the sign of some univariate polynomial R ∈ Q(β)[x] when x =
a1∨x = b1. Fortunately, as both a1 and b1 are rational numbers, the
sign of R(a1) and R(b1) can be decided again using our univariate
sign determination procedure. Hence, evaluating the right-hand side
of Equation 6 requires only arithmetic on rational numbers, and we
can now decide the sign of Q(α, β) with only rational arithmetic
(provided lcx(Q)(β) 6= 0).

To give an example, suppose we want to decide the sign of α−β
when α =

√
2 = (x2 − 2, 1, 2) and β =

√
3 = (x2 − 3, 1, 2).

Figure 1 shows the calculation of

TaQ(x− β, x2 − 2, 1, 2) = −1

provided lcx(x − y)(β) = 1 6= 0. Therefore, we know that
(x− y)[x→

√
2, y →

√
3] is negative.

In Isabelle, we have defined the bivariate sign determination
procedure as bsgn :

definition bsgn_at::"real bpoly ⇒ real ⇒ real
⇒ real" where

"bsgn_at q x y=sgn (bpoly q x y)"

and executability of bsgn_at on the algebraic reals is restored by
the following code equation:

lemma bsgn_at_code2[code]:
fixes q::"real poly poly"

and p1::"int poly" and lb1 ub1::rat
and y::real

shows "bsgn_at q (Alg p1 lb1 ub1) y =
(if valid_alg p1 lb1 ub1
then

(let
q’=degen q y

in (if q’=0 then 0 else
let ps = spmods_y (lift_x p1)

(lift_x (pderiv p1) * q’) y
in changes_bpoly_at ps lb1 y

- changes_bpoly_at ps ub1 y))
else Code.abort (STR ”invalid Alg”)

(λ_. bsgn_at q (Alg p1 lb1 ub1) y))"

where letting q’=degen q y enables q’ to satisfy the assumption
of equation (6), pderiv means derivation and lift_x :: ’a::zero
⇒ ’a poly poly lifts a univariate polynomial to bivariate. More-
over,

changes_bpoly_at ps lb1 y - changes_bpoly_at ps ub1
y

TaQ(x− β, x2 − 2, 1, 2) = Var(SPRemS′(x2 − 2, (2x)(x− y), β)[y → β]; 1, 2)

= Var([x2 − 2, 2x2 − 2xy,−4xy + 8, 64y2 − 128][y → β]; 1, 2)

= Var([x2 − 2, 2x2 − 2xy,−4xy + 8, 64y2 − 128][x→ 1, y → β])

−Var([x2 − 2, 2x2 − 2xy,−4xy + 8, 64y2 − 128][x→ 2, y → β])

= Var([−1,−2y + 2,−4y + 8, 64y2 − 128][y →
√
3])

−Var([2,−4y + 8,−8y + 8, 64y2 − 128][y →
√
3])

= 1− 2 = −1

Figure 1. Deciding the sign of α− β when α =
√
2 and β =

√
3

implements the Var operation. And also, Code.abort throws an
exception when Alg p1 lb1 ub1 fails to be a valid real algebraic
number. Essentially, Lemma bsgn_at_code2 implements (6).

Thanks to bsgn_at, the example in Figure 1 can be executed as

value "bsgn_at [:[:0,-1:],[:1:]:] (Alg [:-2,0,1:] 1 2)
(Alg [:-3,0,1:] 1 2)"

which returns -1.
To restate: we have implemented a decision procedure (called

bsgn_at ) to decide the sign of a bivariate polynomial with rational
coefficients at real algebraic points. This procedure uses no real
algebraic arithmetic, just arithmetic on rational numbers.

3.3 Enable Executability on Algebraic Reals
Although it is possible to do verified algebraic arithmetic as in Coq
[2], with the help of bsgn_at, we can do better. We can actually use
untrusted external code to do such arithmetic, validate the result
and bring it back the framework of higher order logic. The rationale
behind this methodology is that untrusted but sophisticated code
usually offers by far the best performance. Using untrusted code
when building decision procedures improves performance in most
cases; on the other hand, to provide our own trustworthy code would
require costly formal verification. Another benefit of using external
untrusted code is modularity: we can easily substitute one piece of
code by another without modifying any proofs.

The following lemma illustrates the idea of using untrusted code
in algebraic arithmetic:

lemma alg_add_bsgn:
fixes p1 p2 p3::"int poly"

and lb1 lb2 lb3 ub1 ub2 ub3::"rat"
defines "x≡Alg p1 lb1 ub1" and "y≡Alg p2 lb2 ub2"

and "pxy≡[:[:0::real,1:],[:1:]:]"
assumes valid:"valid_alg p3 lb3 ub3"

and bsgn1:"bsgn_at (pcompose (lift_x
(of_int_poly p3)) pxy) x y = 0"

and bsgn2:
"bsgn_at ([:[:-of_rat lb3,1:],[:1:]:]) x y > 0"

and bsgn3:
"bsgn_at ([:[:-of_rat ub3,1:],[:1:]:]) x y < 0"

shows "Alg p3 lb3 ub3 = x + y"

Here, let x = Alg p1 lb1 ub1 and y = Alg p2 lb2 ub2,

• the assumption valid checks if Alg p3 lb3 ub3 is a valid real
algebraic number, which guarantees that p3 has exactly one real
root within interval (lb1, ub1)
• bsgn1 checks if p3 vanishes at x+y, within which pcompose

is polynomial composition and pxy stands for the bivariate
polynomial x+ y

• bsgn2 and bsgn3 checks if lb3 < x+y and x+y < ub3 respec-
tively

With these three assumptions, all of which can be computationally
checked, we can show Alg p3 lb3 ub3 = x + y. Therefore, to
calculate real algebraic addition, we can use untrusted code to
compute p3, lb3 and ub3, and obtain the result as a sound Isabelle
theorem with the help of Lemma alg_add_bsgn.

In order to interact with untrusted code, we have followed the
idea of foreign function interface [12]. First, we declare a constant
alg_add without attaching any definitions (see Figure 2). In Figure 2,
integer × integer encodes rat and integer list encodes int
poly. As Isabelle does not directly link rat to the target language,
we decide to use the quotient of two integers (which appears more
primitive and closer to the native level) to represent rat, and similar
reasons apply to int poly. Also note, in Isabelle, integer is an
equivalent type to int but directly maps to arbitrary precision
integers (e.g. IntInf.int in SML) in the target language when doing
evaluations. Essentially, we let alg_add be an unspecified constant
that takes representations of two algebraic numbers and returns the
representation of their addition and (integer × integer) option,
where (integer × integer) option is a possible optimisation in
case the result is a rational number.

To enable alg_add to do calculations, we use the adaptation
technique to link a constant in Isabelle/HOL to a target language
constant, so that when the logical constant gets called in evaluation,
the target language constant gets invoked instead:

code_printing constant alg_add ⇀ (SML) "untrustedAdd"

where untrustedAdd is currently backed up by Grant Passmore’s
code for algebraic operations in MetiTarski [14]. After such linking,
alg_add becomes executable:

value "alg_add([-2,0,1],(1,1),(2,1))([-3,0,1],(1,1),(2,1))"

evaluates the sum of
√
2 = (x2 − 2, 1, 2) and

√
3 = (x3 −

3, 1, 2), and returns the result ([1, 0, - 10, 0, 1], (2, 1), (4,
1), None), which encodes

√
2 +
√
3 as (x4 − 10x2 + 1, 2, 4).

The code equation for real algebraic addition is the following:

lemma [code]:
"Alg p1 lb1 ub1 + Alg p2 lb2 ub2 =

(let
(ns,(lb3_1,lb3_2),(ub3_1,ub3_2),_)

= alg_add (to_alg_code p1 lb1 ub1)
(to_alg_code p2 lb2 ub2);

(p3,lb3,ub3)
= of_alg_code ns lb3_1 lb3_2 ub3_1 ub3_2

in
(if (*assumptions in the lemma alg_add_bsgn*) then

Alg p3 lb3 ub3
else

Code.abort (STR ”alg_add fails to compute
a valid answer”) (λ_. Alg p1 lb1 ub1
+ Alg p2 lb2 ub2)))"

consts alg_add:: "integer list × (integer × integer) × (integer × integer)
⇒ integer list × (integer × integer) × (integer × integer)
⇒ integer list × (integer × integer) × (integer × integer)
× ((integer × integer) option)"

Figure 2. Logical constant encoding untrusted algebraic addition

where to_alg_code encodes int poly and rat to integer list
and integer ×integer, and of_alg_code does the reverse. The
command Code.abort inserts an exception with an error message,
that is, when our untrusted computation alg_add fails to give a
correct result, an exception will be thrown. This code equation can
be shown to be correct using our previous Lemma alg_add_bsgn.

In a very similar way of exploiting untrusted code, we have
defined subtraction and multiplication. As for negation and inversion,
their code equations do not require untrusted code.

The code equation for negation is the following:

lemma [code]:
"- Alg p lb ub =

if valid_alg p lb ub then
Alg (pcompose p [:0,-1:]) (-ub) (-lb)

else
Code.abort (STR ”invalid Alg”)
(λ_. - Alg p lb ub)"

where pcompose p [:0,-1:] substitutes variable x in a univariate
polynomial p by −x. The rationale behind this code equation is

p(α) = 0 ∧ q(x) = p(−x) =⇒ q(−α) = 0

Also, p(-x) can be shown to have exactly one real root within the
interval (−lb,−ub), provided that p(x) has exactly one within the
interval (lb, ub).

The code equation to invert an algebraic real number is similar:

lemma [code]:
"inverse (Alg p lb ub) =

if valid_alg p lb ub then
(if lb < 0 ∧ 0 < ub then 0
else Alg (rev_poly p) (inverse ub) (inverse lb))

else
Code.abort (STR ”invalid Alg”)

(λ_. inverse (Alg p lb ub))"

where rev_poly simply reverse reverses the coefficients of a polyno-
mial. For example, rev_poly ([:0,1,2:]::int poly) is evaluated
to [:2,1:]. The core idea of this code equation is the same as in the
abstract level (i.e. Equation 2 in Section 2). Note, in valid_alg we
require that lb < ub < 0 or 0 < lb < ub unless Alg p lb ub = 0,
so rev_poly p can be shown to have exactly one real root within
the interval (inverse lb,inverse ub).

By composing multiplicative inverse and multiplication, we
obtain division:

lemma [code]: "Alg p1 lb1 ub1 / Alg p2 lb2 ub2
= Alg p1 lb1 ub1 * (inverse (Alg p2 lb2 ub2))"

As with the comparison operations, we require that the interval
(lb, ub) does not contain 0 unless Alg p lb ub = 0 in valid_alg
p lb ub. Therefore, the sign of an algebraic real can be decided by
the signs of lb or ub :

lemma [code]: "sgn (Alg p lb ub) ==
if valid_alg p lb ub then

if lb > 0 then 1
else if ub < 0 then -1
else 0

else
Code.abort (STR ”invalid Alg”)
(λ_. sgn (Alg p lb ub))"

and the comparison between two algebraic reals can be obtained by
subtraction and compare the result with 0.

Finally, the executability of the arithmetic of our algebraic reals
can be illustrated by the following example:

value "Alg [:-2,0,1:] 1 2 / Alg [:-3,0,1:] 1 2
+ Alg [:-5,0,1:] 2 3 > Alg [:-7,0,2:] 1 2"

which stands for
√
2/
√
3 +
√
5 >

√
7/2 and returns true.

To repeat, we have enabled executable arithmetic and comparison
operations on algebraic reals by deriving new code equations for
the pseudo constructor Alg. Some of these code equations, such as
the one for algebraic addition, depend on untrusted code, whose
results are verified using the bivariate sign determination algorithm
bsgn_at, and thus brought back into higher order logic.

3.4 Linking the Algebraic Reals to the Real Algebraic
Numbers

We have just seen executable arithmetic and ordering operations
on algebraic reals constructed by the constructor Alg, of type int
poly ⇒ rat ⇒ rat ⇒ real. To enable the same executability
on type alg, we only need to build a constructor for alg lifted from
Alg :

lift_definition RAlg:: "int poly ⇒ rat ⇒ rat ⇒ alg" is
"λp lb ub. if valid_alg p lb ub then Alg p lb ub else

0"

and we can then have executable arithmetic and ordering operations
on alg as well:

value "RAlg [:-2,0,1:] 1 2 * RAlg [:-3,0,1:] 1 2
> RAlg [:-5,0,1:] 2 3"

where op * and op > in the command above operate over alg
instead of real.

4. Experiments
This section presents a few examples to demonstrate the efficiency
of our implementation. All the experiments are run on a Intel Core
2 Quad Q9400 (quad core, 2.66 GHz) and 8 gigabytes RAM. When
benchmarking verified operations, the expression to evaluate is
first defined in Isabelle/HOL, and then extracted and evaluated
in Poly/ML. The reason for this is that when invoking value in
Isabelle/HOL to evaluate an expression, a significant and unpre-
dictable amount of time is spent generating code, so we evaluate an
extracted expression to obtain more precise results. The source of
our benchmark is available from the source repository online.

Firstly, we compare evaluations of the same expression using
verified arithmetic from our implementation and unverified ones
from MetiTarski (see Figure 3). The data in Figure 3 indicate that
our verified arithmetic is 2 to 15 times slower than unverified
ones due to overhead in various validity checks and inefficient
data structures. We expect to narrow this gap by further refining
code equations in our implementation. The experiments have also
demonstrated inefficiencies in algebraic arithmetic in the current
version of MetiTarski, which evaluates (

√
2 +
√
6)3 to

(x8−3584x6+860160x4−14680064x2+16777216,
2601

128
,
6125

8
)

Expression Verified evaluation Unverified evaluation (MetiTarki)

(−
√
2) + (−

√
3)− (−

√
5) 0.24s 0.02s

(
∏10

n=2

√
n)(
√
17−

√
19) 0.84s 0.15s∑5

n=2

√
n 1.9s 1.4s

(
√
2 +
√
6)3 1.18s 0.26s

Figure 3. Comparison between verified evaluation and unverified evaluation

while Mathematica8 can evaluate the same expression to

(x4 − 3328x2 + 4096, 2, 59)

instantly. By basing our untrusted code on more sophisticated
algebraic arithmetic implementations such as Z3 and Mathematica,
which effectively control coefficient and degree growth, we should
obtain further improvements in our algebraic arithmetic.

We have also experimented with our bivariate sign determination
procedure alone, which appears to be quite efficient. For example,
given the large bivariate polynomial P (x, y) shown in Figure 4,
our bsgn_at can decide P (

√
6,
√
7) = 0 or P (

√
13,
√
29) > 0

in less than 0.05s. Note, our current bsgn_at always calculates a
remainder sequence no matter whether the result is −1, 0 or 1, so
bsgn_at should take similar amount of time if the input argument is
of similar complexity. In the future, we may optimize bsgn_at by
letting it attempt to decide the sign using interval arithmetic before
calculating a remainder sequence; in this case bsgn_at may run
much faster if the polynomial does not vanish at the algebraic point.

5. Related Work
The most closely related work is Cyril Cohen’s construction of the
real algebraic numbers in Coq [2], from which we have gained a lot
of inspiration. There are some major differences between our work
and his:

• Cohen’s work is part of the gigantic formalization of the odd
order theorem [5] and is mainly of theoretical interest. Our work,
on the contrary, is for practical purposes, as we are intending
to build effective decision procedures on the top of our current
formalization. This difference in intent is fundamental and leads
to different design choices, such as whether to use efficient
untrusted code.
• Our formalization follows Isabelle’s tradition of separating ab-

straction and implementation, that is, formalizing theories first
and restoring executability afterwards. It is possible to switch
to another encoding of real algebraic numbers (such as Thom
encoding) without modifying any definition or lemma on the ab-
stract level. It is also possible to have multiple implementations
of one abstraction [7], so that when doing proof by reflection
the code generator can choose the most efficient one depending
on the situation. On the other hand, Cohen’s formalization is
constructive and therefore should be executable, though it may
not be very efficient.
• In Cohen’s formalization, arithmetic on real algebraic numbers

is defined via verified bivariate resultants, while ours is mainly
based on a bivariate sign determination procedure and some
untrusted code.

8 We use the RootReduce and IsolatingInterval command in Mathematica 9
to find the defining polynomial and root isolation interval.

6. Discussion
6.1 Modularity
The dependencies between the parts of our formalization are shown
in Figure 5. Modularity in our formalization is reflected in two ways:

• Separation between the abstract type, alg, and the finite encod-
ing, Alg. Switching to another encoding does not affect anything
on the abstract level or further theories based on the abstraction.
• Use of untrusted code. Untrusted code is outside the logic of

Isabelle/HOL (which is why we have used a dashed arrow in
Figure 5 to indicate the detached relation), hence we do not need
to modify our formalization as we revise the untrusted code, or
substitute new code.

This modularity should make our formalization easier to maintain.

6.2 A Potential Problem
There is one potential drawback with our formalization, and it is
related to the use of untrusted code. Recall that when interfacing
with untrusted code, we declare a constant in higher order logic
without specifying it and link it to a constant in the target language.
In this case the logic constant can be executed but no lemmas are
associated with it. However, this method may undermine proofs
through reflection unless referential transparency9 is guaranteed in
the target language constant. For example, consider the ML function
serial, which maintains a counter and returns the number of times
it is called. Linking an Isabelle constant, say time, to the target
language constant serial breaks referential transparency:

consts time :: "unit ⇒ integer"

code_printing constant time ⇀ (SML) "serial"

we have

value "time () = time ()"

which returns false and breaks reflexivity. This example is due to
Lochbihler and Züst [12]. So we can see that any use of external
code potentially makes the system inconsistent. In the short term,
this is something we have to live with.

Note, our bivariate sign determination procedure (bsgn_at ) does
not depend on any untrusted code (as shown in Figure 5), hence this
problem does not apply to bsgn_at.

6.3 Future work
Here are some possible extensions to our current formalization:

• to improve the efficiency of our untrusted code. The efficiency
of our algebraic arithmetic critically depends on the underlying
untrusted code, and dramatic improvements in efficiency can be
expected if the untrusted part is optimized. And thanks to our
modularity, we do not need to modify existing formalizations to
accommodate changes in the untrusted part.

9 Programs always return the same value and have the same effect if they are
given the same input.

P (x, y) = y14x24 − 49y12x24 + 1029y10x24 − 12005y8x24 + 84035y6x24 − 352947y4x24 + 823543y2x24 − 823543x24 +

4y15x23 − 196y13x23 + 4116y11x23 − 48020y9x23 + 336140y7x23 − 1411788y5x23 + 3294172y3x23 − 3294172yx23 + 6y16x22 −
380y14x22 + 10388y12x22 − 160524y10x22 + 1536640y8x22 − 9344692y6x22 + 35294700y4x22 − 75765956y2x22 + 70824698x22 +
4y17x21 − 488y15x21 + 18424y13x21 − 348488y11x21 + 3841600y9x21 − 25950008y7x21 + 106354696y5x21 − 243768728y3x21 +
240474556yx21 + y18x20 − 435y16x20 + 23124y14x20 − 565068y12x20 + 7991214y10x20 − 70978362y8x20 + 404376420y6x20 −
1441435548y4x20 + 2937577881y2x20 − 2619690283x20 − 240y17x19 + 21360y15x19 − 717360y13x19 + 12759600y11x19 −
135416400y9x19 + 891443280y7x19 − 3585941520y5x19 + 8103663120y3x19 − 7906012800yx19 − 60y18x18 + 14220y16x18 −
682560y14x18 + 15664320y12x18 − 210533400y10x18 + 1786632120y8x18 − 9753438240y6x18 + 33374668320y4x18 −
65372843340y2x18 +56083278300x18 +6480y17x17− 505440y15x17 +15876000y13x17− 271162080y11x17 +2800526400y9x17−
18078953760y7x17 + 71662358880y5x17 − 160096759200y3x17 + 154760200560yx17 + 1620y18x16 − 277020y16x16 +
12299040y14x16 − 269256960y12x16 + 3483988200y10x16 − 28557590040y8x16 + 150730554240y6x16 − 498587050080y4x16 +
943236739620y2x16 − 780471701100x16 − 103680y17x15 + 7516800y15x15 − 226074240y13x15 + 3751816320y11x15 −
37962691200y9x15 + 241343141760y7x15 − 945333244800y5x15 + 2091930986880y3x15 − 2006546048640yx15 −
25920y18x14 + 3576960y16x14 − 148770432y14x14 + 3128969088y12x14 − 39196700928y10x14 + 311791939200y8x14 −
1597046855040y6x14 + 5119436939904y4x14 − 9362458478016y2x14 + 7462643602176x14 + 1088640y17x13 − 75333888y15x13 +
2197746432y13x13 − 35697376512y11x13 + 355480151040y9x13 − 2232206242560y7x13 + 8658032737536y5x13 −
19006687252224y3x13 + 18110145400704yx13 + 272160y18x12 − 32169312y16x12 + 1263911040y14x12 − 25636818816y12x12 +
311754598848y10x12 − 2411253230400y8x12 + 11999023392384y6x12 − 37270525541760y4x12 + 65761344808992y2x12 −
50251170777696x12 − 7838208y17x11 +525159936y15x11 − 14978815488y13x11 +239276975616y11x11 − 2352442176000y9x11 +
14622780566016y7x11 − 56251597312512y5x11 + 122646925287936y3x11 − 116191823956992yx11 − 1959552y18x10 +
205752960y16x10 − 7683683328y14x10 + 150666034176y12x10 − 1780750718208y10x10 + 13398166381824y8x10 −
64739208683520y6x10 + 194443460499456y4x10 − 329440707211392y2x10 + 239069288578176x10 + 39191040y17x9 −
2564213760y15x9 + 71876367360y13x9 − 1133012966400y11x9 + 11022871910400y9x9 − 67938530042880y7x9 +
259521420856320y5x9 − 562515973125120y3x9 + 530240466470400yx9 + 9797760y18x8 − 936385920y16x8 +
33399164160y14x8 − 634130622720y12x8 + 7287769843200y10x8 − 53313060971520y8x8 + 249688212560640y6x8 −
722246796875520y4x8 + 1165376329568640y2x8 − 789597216374400x8 − 134369280y17x7 + 8633226240y15x7 −
238673433600y13x7 + 3721659540480y11x7 − 35891546342400y9x7 + 219624001551360y7x7 − 833893702548480y5x7 +
1798206799334400y3x7 − 1687547919375360yx7 − 33592320y18x6 + 2972920320y16x6 − 101683952640y14x6 +
1871528924160y12x6 − 20912730854400y10x6 + 148566805309440y8x6 − 672422071587840y6x6 + 1861440445025280y4x6 −
2821801438955520y2x6 + 1729044999360000x6 + 302330880y17x5 − 19147622400y15x5 + 523435530240y13x5 −
8088560363520y11x5 + 77428953907200y9x5 − 470864825948160y7x5 + 1778446285056000y5x5 − 3817731358586880y3x5 +
3568748878679040yx5 + 75582720y18x4 − 6273365760y16x4 + 206451680256y14x4 − 3686763838464y12x4 +
40038373799424y10x4 − 275719665553920y8x4 + 1200874682004480y6x4 − 3151406817119232y4x4 + 4386241354376448y2x4 −
2269890275159808x4 − 403107840y17x3 + 25234550784y15x3 − 683429031936y13x3 + 10480561975296y11x3 −
99689778155520y9x3 + 602977978552320y7x3 − 2266926198018048y5x3 + 4846858942205952y3x3 − 4514882302328832yx3 −
100776960y18x2 + 7921069056y16x2 − 251539292160y14x2 + 4361304342528y12x2 − 46001094580224y10x2 +
306328298895360y8x2 − 1276416305160192y6x2 + 3130065461698560y4x2 − 3834330190580736y2x2 + 1377703055490048x2 +
241864704y17x− 14995611648y15x+ 402946596864y13x− 6139009916928y11x+ 58071715430400y9x− 349591726891008y7x+
1308936465801216y5x− 2788603774967808y3x+ 2589417791041536yx+ 60466176y18 − 4534963200y16 + 139314069504y14 −
2346571358208y12 + 24016802310144y10 − 154180404467712y8 + 609753012019200y6 − 1365846746923008y4 +
1344505391502336y2 − 49796495981568

Figure 4. A large bivariate polynomial

Abstract type alg
Pseudo constructor Alg

and bivariate sign
determination bsgn

Untrusted code

Algebraic arithmetic on Alg

Executable real algebraic number alg

Figure 5. Dependence tree of our formalization of real algebraic numbers

• to generalize the bivariate sign determination procedure to
decide the sign of a multivariate polynomial with rational
coefficients. The idea behind a bivariate and a multivariate
procedure should be the same, and the only reason we did not
build a multivariate sign determination procedure directly is that
Isabelle’s multivariate polynomial library [7] is not finished yet.
• to integrate the sign determination algorithm with sophisticated

interval arithmetic: to decide the sign using interval arithmetic
first (could refine the interval for algebraic numbers a couple of
times before giving up) and revert to the current signed remain-
der sequences if fails, as others have done [3, 15]. Moreover,
dyadic rationals10 (numbers of the shape n2m for n, m ∈ Z)
can be used to improve performance with intervals.
• to improve arithmetic between real algebraic numbers and ra-

tional numbers. For example, given a real algebraic number
α = (p, lb, ub) and a rational number r, the defining polyno-
mial for α + r is p(x − r), which merely needs polynomial
composition instead of calculating resultants. However, the in-
terval (lb+ r, ub+ r) may need to be refined to exclude zero,
and the termination of such a refinement function may take
some effort to show. For now, we treat r as an algebraic number
(x − r, r/2, 2r) and deploy algebraic arithmetic, which is not
very efficient. And real algebraic numbers that are also rational,
such as (x2 − 4, 1, 3), should be converted to rational numbers.

7. Conclusions
In this paper, we have formalized real algebraic numbers in Is-
abelle/HOL. The formalization is on two levels:

• on the abstract level, proofs in abstract algebra are used to show
that real algebraic numbers, which are formalized as a subset of
real numbers, form an ordered field;
• on the implementation level, an additional pseudo constructor

for real numbers and related code equations are proved via a
bivariate sign determination procedure and some untrusted code.

Experiments indicate that overhead in our verified algebraic arith-
metic is reasonable (compared to unverified ones) and the bivariate
sign determination procedure alone is quite efficient already.

When building practical decision procedures involving real
algebraic numbers, users of our formalization should first try to
build the procedure upon our sign determination procedure, as it
only uses rational arithmetic and is much more efficient than exact
algebraic arithmetic.

[9] B. Huffman and O. Kunčar. Lifting and Transfer: A modular design for
quotients in Isabelle/HOL. In Certified Programs and Proofs, pages
131–146. Springer, 2013.

[10] W. Li. The Sturm-Tarski Theorem. Archive of Formal Proofs, Sept.
2014. ISSN 2150-914x. http://afp.sf.net/entries/Sturm_
Tarski.shtml, Formal proof development.

[11] W. Li, G. O. Passmore, and L. C. Paulson. A complete decision
procedure for univariate polynomial problems in Isabelle/HOL. arXiv
preprint arXiv:1506.08238, 2015.

[12] A. Lochbihler and M. Züst. Programming TLS in Isabelle/HOL. In
Isabelle Workshop, volume 2014, 2014.

[13] T. Nipkow, L. C. Paulson, and M. Wenzel. Isabelle/HOL: A Proof
Assistant for Higher-Order Logic. Springer, 2002.

[14] G. O. Passmore, L. C. Paulson, and L. de Moura. Real algebraic
strategies for MetiTarski proofs. In Intelligent Computer Mathematics,
pages 358–370. Springer, 2012.

